
1. Что такое СОМ-технология
Все поломки случаются в тот день, когда у техника выходной.

Из законов Мэрфи

При решении многих задач часто можно обнаружить, что "все уже сделано за нас": существуют программы, выполняющие необходимые нам функции. Например, если вашей программе надо выводить данные "в красивом виде" и с возможностью печати, то с такой задачей прекрасно справляется Microsoft Word. Если необходимо выводить график или векторное изображение, то для этого прекрасно подходит AutoCAD и т.д. Программирование в своей задаче вывода текста и графики на печать или своего векторного редактора – дело трудоемкое и неблагодарное, а главное, бессмысленное: все равно лучше, чем в Word и AutoCAD, вряд ли получится. 
Для использования мощных возможностей "чужих" программ в собственных разработках в ОС Windows применяется так называемая COM-технология. COM означает Common Object Interface (общий интерфейс объектов). Идея COM-технологии состоит в том, что ваша программа может импортировать так называемую библиотеку типов внешнего приложения и получить доступ к используемым в этом приложении объектам, их свойствам и методам. Вызывая на выполнение из своей программы (СОМ-клиента) методы объектов другой программы (СОМ-сервера), можно выполнять самые разнообразные функции. При этом сам СОМ-сервер в явном виде не запускается: можно сформировать сложный документ в Word и сохранить его в файле совершенно незаметно для пользователя. 

Рис.  1 – Создание приложения по СОМ-технологии.

СОМ-серверами являются большинство распространенных программ: все компоненты пакета Microsoft Office всех версий, AutoCAD, Компас, Visio, SolidWorks… Таким образом, создатели этих программ предусматривают использование функциональности, заложенной в их творения, сторонними разработчиками. Осталось научиться это делать. 
2. Вывод отчета при помощи Microsoft Word
  У бюрократов поток бумаг тем обильнее, чем больше времени 
тратится на отчеты о все меньшем объеме работы.
Из законов Мэрфи

Рассмотрим использование СОМ-технологии на примере. Пусть в результате работы нашей программы вычисляется среднее арифметическое матрицы вещественных чисел 5×6 элементов. Мы хотим получить результаты работы в виде красивого отчета (doc-файла) примерно следующего вида (Рис.  2).

Рис.  2 – Ожидаемый вид отчета. 
Стандартными средствами Delphi получить такой документ, пригодный к распечатке, весьма затруднительно. Пришлось бы отрисовывать его вручную, по клеточке, на объекте Tcanvas и потом развлекаться весьма трудоемким программированием взаимодействия вашей программы с очередью печати Windows. Мы пойдем другим путем – переложим всю трудоемкую работу по форматированию на Microsoft Word. В результате совместных усилий нашей программы и Word мы получим doc-файл на диске, который счастливый пользователь сможет загрузить в тот же Word, просмотреть, распечатать и т.д. 
2.1.  Проверка наличия СОМ-сервера на компьютере

Вы будете смеяться, но предположение о том, что Word установлен на 100% компьютеров, неверно. Всегда есть риск натолкнуться на любителя извращений, который пользуется WordPerfeсt, Лексиконом или даже WordPad. Поэтому первым делом надо проверить, установлен ли Word на компьютере. А после этого перед обращением к СОМ-серверу нужно проверить, не запущен ли Word, и, если это так, настоятельно посоветовать пользователю закрыть его. Дело в том, что возможна ситуация, когда Word будет получать команды одновременно от вашей программы и от пользователя, нажимающего кнопочки на клавиатуре. Такого издевательства Word обычно не выдерживает и с горя вешается. 
ОБЩЕЕ ПРАВИЛО: при работе с любым СОМ-сервером запретите пользователю им пользоваться, пока с СОМ-сервером работает ваша программа.

Чтобы проверить факт наличия на компьютере Microsoft Word, надо заглянуть в системный реестр Windows и выяснить, если там запись  'Word.Application' в разделе HKEY_CLASSES_ROOT. Для  доступа к системному реестру в Delphi предусмотрен специальный тип данных Tregistry:
function IsWordInstalled: Boolean;

var

  Reg: TRegistry;

begin 

  Reg := TRegistry.Create; 

  Reg.RootKey := HKEY_CLASSES_ROOT; 

  Result := Reg.KeyExists('Word.Application'); 

  Reg.Free
end;
Теперь проверим, не запущен ли Word в данный момент и при необходимости предложим пользователю закрыть его. Делается это так: 

FUNCTION WordClosed:BOOLEAN;

function FindNextWnd(StartHWND: HWND; AString : String): HWND;

var

  Buffer : array [0..255] of char;

  s:STRING;

  i:BYTE;

begin

 FillChar(Buffer,256,#0);

  Result := StartHWND;

  repeat

    Result := FindWindowEx(0, Result, nil, nil);

    GetWindowText(Result, Buffer, SizeOf(Buffer));

    i:=0;

    s:='';

    WHILE (i<=255) AND (buffer[i]<>#0) DO

     BEGIN

      s:=s+buffer[i];

      INC(i)

     END;

   IF s<>'' THEN

    if Pos(ANSIUpperCase(AString),ANSIUpperCase(s))<>0 then

     Break;

  until (Result = 0);

end;

VAR h:HWND; r:INTEGER;

CONST MSW='Microsoft Word';

BEGIN

 r:=mrRetry;

 REPEAT

  h:=FindNextWnd(Self.Handle,MSW);

  IF h<>0 THEN

   r:=MessageDlg('Завершите работу '+MSW,mtConfirmation,[mbRetry, mbCancel],0)

 UNTIL (h=0) OR (r=mrCancel);

 Result:=r<>mrCancel

END;
Функция FindNextWnd перебирает заголовки всех открытых в Windows окон и ищет, есть ли среди них окно с текстом "Microsoft Word". Если запустить Word, то можно увидеть, что в заголовке его окна эти слова всегда присутствуют (Рис.  3). 

[image: image2.png]3]

@ain [paexa Bna Boraska OopusT Cepenc T

Ofiwresi +Tlep + TimesNewRoman  » 12+

CONST MSW="Mic




Рис.  3 – Заголовок окна Word.
Внутри цикла REPEAT…UNTIL проверяется, запущен ли Word, и выводится сообщение пользователю "Завершите работу Microsoft Word" с двумя вариантами ответа: "Повторить" (при этом пользователь должен переключиться в Word и завершить его работу) и "Отмена". Вся функция WordClosed возвращает True, если Word закрыт, и False, если пользователь нажал кнопку "Отмена". 
При такой проверке запуска Word возможна ситуация, когда на панели задач Word'а явно нет, а функция WordClosed упорно сообщает, что Word запущен. Дело в том, что Word виснет не менее часто, чем остальные программы, оставаясь при этом в списке процессов Widows в "полумертвом" состоянии, где его и находит наша функция. В таком случае придется нажать Ctrl+Alt+Del, найти в списке процессов WINWORD.EXE и завершить зависший процесс вручную (Рис.  4). 
3. Подключение к СОМ-серверу Word из Delphi
Самая грубая ошибка будет выявлена, лишь когда программа 
пробудет в производстве по крайней мере полгода.
Из законов Мэрфи

Разработчики Delphi позаботились о том, чтобы сделать удобной работу по СОМ-технологии с наиболее часто встречающимися программами, и Word, разумеется, входит в это число. При установке Delphi 6 инсталлятор задает вопрос, для какой версии Word (97 или 2000) устанавливать компоненты связи с СОМ-сервером. Принципиальной разницы между версиями Word с точки зрения СОМ-технологии нет: все, что работает в 97-й версии, будет работать и в 2000, и в XP (Delphi 7 имеет специальные компоненты для Word XP).  

[image: image3.png]= A=
ofin DopereTpn Bia Sapepuene paboTer Crpasica

Mpnoxcesya MPoueccs!  Buictpoeficrene | Cets

Vi obpasa Vimnonsosat... U1 Mawam A
taskmar.exe Aa 00 4552K6
Far.exe a 00 46s4K5
explorer.exe ama 0 1932K5
ibserver.exe SvsTEM o 3172K8
WINWORD.EXE  mia 00 34308K5
outpost.exe SYSTEM 0 120805
nusvez.exe SYSTEM 0 1740KE
bguard.ve. SYSTEM o 20528
CDACHIBAEXE  SYSTEM o sleKE
slg.exe LOCALSERVICE 00 3572K6
LEAPPS.EXE SYSTEM 0 2764KE
spoolsv.xe SYSTEM 0 5136KE
LEXBCES EXE SYSTEM 0 2840KE
svchost.exe LOCALSERVICE 00 3392KG
svchost exe NETWORK SERVICE 00 2808KE
Photoshop.exe  Auma 0 61712K5
svchost.exe SYsTEM o s2KE
bt e avaTem M amo

Qrofipaars npouecce scex nonssosaTene

asepunTs npolece

Mpoueccos: 34 3arpysxali: 2% Beiaenenwis nanatv: 26310 6

<



 
Рис.  4 – Зависший Word в списке процессов. 
Для подключения к основным СОМ-серверам на закладке "Servers" палитры компонентов имеются компоненты для Word, Excel, PowerPoint, Access, Outlook,  и т.д. (Рис.  5).

Рис.  5 – Страница Servers палитры компонентов в Delphi.
Следует отметить, что для подключения к СОМ-серверу вовсе необязательно использовать предлагаемые компоненты. Скажем, для работы с AutoCAD в Delphi нет соответствующего компонента, но это нисколько не помешает нам подключиться к нему (см. п. 7).
Поместим на форму невизуальные компоненты WordApplication, WordDocument, WordFont и WordParagraphFormat. Их функции представлены на Рис.  6. Компонент WordApplication обеспечивает собственно подключение к Word, открытие и сохранение документов. Компонентов типа WordDocument может быть несколько, каждый из них соответствует одному документу, открытому в Word. Компоненты WordFont и WordParagraphFormat отвечают за форматирование текущего абзаца и текущего шрифта в документе. 
Для задания параметров, передаваемых непосредственно методам СОМ-сервера, используется специальный тип данных OLEVariant. Если мы работаем только с компонентами вроде WordFont и WordParagraphFormat, тип данных OLEVariant не нужен. Но при прямом вызове методов объектов Word без него не обойтись. 

Рис.  6 – Функции компонентов управления Word'ом.
Поэтому заведем в нашей процедуре переменную par типа OLEVariant. Далее подключаем компонент WordApplication1 к серверу вызовом его метода Connect, создаем новый пустой документ, добавив элемент список  документов (WordApplication1.Documents.Add) и присвоив ссылку на новый документ компоненту WordDocument1. Встроенные константы EmptyPar используются для обозначения пустого параметра – в рядке случаев метод объекта COM-сервера требует указания множества параметров, которые нам не нужны, и вместо них можно просто написать EmptyPar.
Добавление текста, таблиц, графики в документ выполняется не совсем обычным образом. В Word применено понятие выделенного фрагмента (selection), сокращенно ВФ. ВФ – это или выделенная блоком часть документа, если такое выделение присутствует, или положение текстового курсора в документе (так называемая точка вставки). В нашем случае ВФ – это положение курсора. Объект Selection относится к компоненту WordApplication, а не WordDocument, так как вводить текст можно только в один активный документ.
Простейший способ добавить текст – использовать метод WordApplication1.selection.InsertAfter(текст). Переход на новую строчку выражается символом с кодом 13. Далее документ сохраняется в файл с указанным именем, Word закрывается и наша программа отключается от СОМ-сервера.
Следует иметь в виду, что работа с СОМ-сервером – дело небыстрое. Вывод даже простого отчета может занять несколько десятков секунд. Поэтому желательно информировать пользователя о ходе процесса и о его завершении. 
procedure TForm1.Button1Click(Sender: TObject);

var par:OLEVariant;

begin

 // подключение к Word

 WordApplication1.Connect;

 // Создание нового документа и подключение к нему WordDocument1  WordDocument1.ConnectTo(WordApplication1.Documents.Add
(EmptyParam,EmptyParam,EmptyParam,EmptyParam));

 // Добавление текста в Word
 WordApplication1.selection.InsertAfter('Результаты работы гениальной  программы студента Б. Гейтса'+#13);
 par:=ExtractFilePath(Application.ExeName)+'out.doc';

 // Сохранение документа

 WordDocument1.SaveAs(par);

 // Закрытие документа в Word
 WordDocument1.Close;

 // Завершение работы с документом

 WordDocument1.Disconnect;

 // Закрытие Word
 WordApplication1.Quit;

 // Завершение сеанса работы с Word
 WordApplication1.Disconnect;

 MessageDlg('Вывод отчета завершен',mtInformation,[mbOK],0)

end;
4. Управление форматированием документа

После того, как кожух закрепили 96 удерживающими болтами, 

окажется, что внутрь забыли положить прокладку.

Из законов Мэрфи.

Попробуем сделать наш отчет красивым. Для этого мы можем управлять форматированием абзацев (выравнивание, межстрочный интервал, отступ первой строки) и форматированием шрифта (гарнитура шрифта, размер, жирный-курсив-подчеркнутый и т.д.) Заданное форматирование действует на текущий абзац. Обычно схема работы такова:
- вывести в Word текст абзаца;

- применить к нему нужное форматирование;

- вывести в Word текст следующего абзаца;

- применить к нему форматирование

и т.д. 

Для задания параметров абзаца служит компонент WordParagraphFormat. Свойств у него много, поэтому рассмотрим только главные:
	Свойство
	Что делает
	Возможные значения

	Alignment
	Способ выравнивания текста в абзаце
	wdAlignParagraphCenter – по центру
wdAlignParagraphJustify – по ширине
wdAlignParagraphLeft - влево
wdAlignParagraphRight - вправо

	FirstLineIndent
	Отступ первой строки в пунктах 
(1 см = 28.35 пункта)
	Целое число от 0 и выше

	LineSpacing
	Межстрочный интервал в пунктах
	Значение 12 соответствует 1 интервалу, 18 – 1.5, 24 – 2-м. 


Стандартный межстрочный интервал также удобно задавать вызовами методов Space1, Space15, Space2.
Шрифтом управляет компонент WordFont. Вот его основные свойства:
	Свойство
	Что делает
	Возможные значения

	Name
	Название гарнитуры как текстовая строка
	'Times', 'Arial', 'Courier'…

	Size
	Размер шрифта в пунктах
	Целое число от 6 и выше. Стандартный размер шрифта – 14

	Italics 
	вкл/выкл курсив
	Целое число 0 или 1

	Underline
	вкл/выкл подчеркивание
	Целое число 0 или 1

	Bold
	вкл/выкл жирный шрифт
	Целое число 0 или 1

	Subscript
	вкл/выкл верхние индексы
	Целое число 0 или 1

	Superscript
	вкл/выкл нижние индексы
	Целое число 0 или 1

	Color
	цвет шрифта
	Обычные константы цвета: clRed, clBlue…


Рассмотрим использование перечисленных свойств в программе. В начале один раз надо подключить компоненты WordFont1 и WordParagraphFormat1 к текущему абзацу и шрифту при помощи их метода ConnectTo. 
…

// Добавление текста в Word

 WordApplication1.selection.InsertAfter

('Результаты работы гениальной программы студента Б. Гейтса'+#13);

 // подключение к текущему шрифту
 WordFont1.ConnectTo(WordApplication1.Selection.Font);

// подключение к текущему абзацу
WordParagraphFormat1.ConnectTo

(WordApplication1.Selection.ParagraphFormat);

 // название шрифта
 WordFont1.Name:='Arial';

 // размер

 WordFont1.Size:=14;

// выравнивание по центру

 WordParagraphFormat1.Alignment:=wdAlignParagraphCenter;

 // Установка нулевого абзацного отступа

 WordParagraphFormat1.FirstLineIndent:=0;

 // Снимаем выделение текста

 par:=wdCollapseEnd;

 WordApplication1.Selection.Collapse(par);
WordApplication1.selection.InsertAfter

('Второй абзац'+#13);

Особую важность играет вызов метода Collapse. Этот метод позволяет перейти к выводу в Word следующего абзаца, у которого, вероятно, будет иное оформление. Если забыть выполнить метод Collapse, то вновь вводимый текст заменит ранее введенный. В приводимом примере строчка "Результаты работы гениальной программы студента Б. Гейтса" просто исчезла бы из отчета.

5. Работа с таблицами

Не позволяйте компьютеру догадаться, что вы спешите.
Из законов Мэрфи
Давайте научимся вставлять таблицы в документ Word. Каждая таблица в программе будет представлена переменной типа Table. Поэтому прежде всего заведем такую переменную:
VAR t:Table;

При вставке новой таблицы в документ надо указать количество строк и столбцов в ней. Делается это так: 

t:=WordDocument1.Tables.Add(WordApplication1.Selection.Range,

6,5,EmptyParam,EmptyParam);
Здесь 6 – число строк, а 5 – число столбцов. Не забудьте присвоить ссылку на новую таблицу переменной t.

Новой таблице желательно придать приличный вид. Чтобы линии сетки таблицы были видны, задаем так называемый автоформат таблицы. Здесь уже понадобятся две переменные типа OLEVariant: par и par1.
par:=wdTableFormatGrid1;

par1:=wdTableFormatApplyBorders;

t.AutoFormat(par, par1, EmptyParam, EmptyParam, EmptyParam, EmptyParam, EmptyParam, EmptyParam, EmptyParam, EmptyParam);

Теперь давайте разместим таблицу по центру страницы:

t.Rows.Alignment:=wdAlignRowCenter;
Наконец, очень удобно включить функцию автоподбора ширины столбцов, что позволяет не задумываться о подгонке ширины текста в ячейках: 
t.AllowAutoFit:=True;

par:=wdAutoFitContent;

t.AutoFitBehavior(par);
Для доступа к тексту внутри конкретной ячейки таблицы используется ее свойство Cell(строка, столбец). Например, вот так можно заполнить нашу таблицу случайными числами (не забудьте выделить память под целые переменные i,j и вещественную a):  
// цикл по ячейкам таблицы
 for i:=1 to 6 do

  for j:=1 to 5 do

   begin

    // занесение значения в ячейку с координатами i,j

    t.Cell(i,j).Select;

    a:=RANDOM(100);

    WordApplication1.selection.InsertAfter

    (FloatToStrF(a,ffFixed,10,4))

   end;
Самое трудное при работе с таблицей – это выбраться из нее, то есть перенести точку ввода текста и текстовый курсор за пределы таблицы, чтобы продолжить писать текст после нее. Делается это так (предполагаем, что перед этим мы выводили текст в последнюю ячейку таблицы и курсор находится в ней): 

// Переводим курсор за пределы таблицы

 par:=wdStory;

 par1:=wdMove;

 WordApplication1.Selection.EndKey(par,par1);

Метод EndKey имитирует нажатие клавиш Ctrl+End, т.е. переход в самый конец документа, под таблицу. 
6. Запуск Word из внешней программы
Общая сумма разума на планете - величина постоянная, а население растет...
Из законов Мэрфи

Наша программа, работая как СОМ-клиент, заставляет Word сформировать на диске doc-файл с указанным именем. Часто желательно, чтобы по завершении генерации этот файл оставался бы загруженным в Word, чтобы пользователь тут же смог его просмотреть. Простейший способ – выбросить из программы методы WordDocument1.Close и WordApplication1.Quit. Тогда созданный нами документ и весь Word не будут закрываться. 
Способ более интересный – запуск Word в любой момент из нашей программы и загрузка в него файла с указанным именем. Причем так можно запускать не только Word, но и любую другую программу, связанную с определенным типом файлов (скажем, для dwg-файлов запустится AutoCAD). Делается это так:
ShellExecute(0, 'OPEN', pchar(имя файла), nil, nil, SW_MAXIMIZE)

Не забывайте, что, если в имени файла содержатся пробелы, то его надо взять в кавычки. Параметр SW_MAXIMIZE раскрывает окно Word'а на весь экран. 
7. Работа с AutoCAD по COM-технологии

Когда корабль тонет, следует держаться поближе 
к шлюпкам, а не к капитанскому мостику.
Из законов Мэрфи

Рассмотрим теперь работу с СОМ-сервером, который Delphi не знаком. Для примера возьмем AutoCAD. Прежде всего в проект необходимо импортировать библиотеку типов – Delphi должен узнать, какие методы и свойства есть в искомом COM-сервере. Создаем новый проект, в меню Project выбираем пункт Import Type Library. Открывается окно импорта библиотеки типов (Рис.  7).

Рис.  7 – Окно импорта библиотеки типов.
В списке в верхней части окна перечислены все установленные на компьютере СОМ-серверы. В этот список надо добавить библиотеку типов AutoCAD. Для версии 2004 она живет в файле acax16enu.tlb, который находится в директории Program Files\Common Files\Autodesk Shared. Щелкаем по кнопке Add… и открываем файл acax16enu.tlb. В раскрывающемся списке Palette Page можно выбрать ту страницу палитры компонентов, на которую будут добавлены новые компоненты для работы с AutoCAD. Можно выбрать страницу Servers или предварительно создать отдельную страницу (щелчок правой кнопкой мыши над палитрой). 
Указав файл и страницу, щелкаем по кнопке Install. Появляется окно Install  (Рис.  8). В нем просто щелкаем по кнопке ОК.

[image: image7.emf]
Рис.  8 – Окно установки компонентов.

Затем выводится окно компиляции (Рис.  9). Щелкаем по кнопке Compile. В процессе компиляции обычно на одной-двух строчках выскакивают ошибка – эти ошибки допущены разработчиками AutoCAD. Просто закомментируйте эти строчки. После успешной компиляции щелкните по кнопке Install. В результате в палитре компонентов на выбранной странице будет добавлен ряд новых компонентов, отвечающих за связь с AutoCAD. 

Рис.  9 – Окно компиляции.

Закрываем проект, создаем новый проект и помещаем на форму компонент AcadDocument и кнопку. В обработчике нажатия кнопки подключаемся к AutoCAD и проводим отрезок из точки (0;0;0) в точку (10;10;0):  

procedure TForm1.Button1Click(Sender: TObject);

var

  v : OleVariant;

  BeginPoint, EndPoint: OleVariant;

  Line: IAcadLine;

  par,par1:OLEVariant;

begin

  Try

   // пытаемся присоединиться к уже запущенной копии AutoCad

   v := GetActiveOleObject('AutoCAD.Application');

AcadDocument1.ConnectTo((IDispatch(v) as     IAcadApplication).ActiveDocument);

  Except

// если не получилось, то запускаем новую копию AutoCad

    AcadDocument1.Connect

  end;

// задаём массив из трёх вещественных чисел

BeginPoint:= VarArrayCreate([0, 2], varDouble);

EndPoint:=   VarArrayCreate([0, 2], varDouble);

BeginPoint[0]:= 0.0;  // координата X

BeginPoint[1]:= 0.0;  // координата Y

BeginPoint[2]:= 0.0;  // координата Z

EndPoint[0]:= 10.0;  // координата X

EndPoint[1]:= 10.0;  // координата Y

EndPoint[2]:= 0.0;  // координата Z

  // функция AddLine добавляет отрезок в чертеж
  Line:= AcadDocument1.ModelSpace.AddLine(BeginPoint, EndPoint);

  // поменять цвет отрезка
  Line.color:= acRed;

  // поменять толщину линии

  Line.Lineweight:= acLnWt040;
  // сохраним файл
  par:=ExtractFilePath(Application.ExeName)+'1.dwg';

  par1:=ac2004_dwg;

  AcadDocument1.SaveAs(par,par1);

  AcadDocument1.Close;

  AcadDocument1.Disconnect;

  MessageDlg('Вывод в AutoCAD завершен', mtInformation, [mbOK],0)

end;
Цвет и толщина линии задаются специальными константами вида acRed, acLnWt40 и т.д. Полный список этих констант можно найти в файле AutoCAD_TLB.pas, который был создан при подключении библиотеки типов и находится в директории Delphi\Imports. 
Для упрощения кода программы введем функцию GetPoint, которая преобразует три вещественных числа (координаты точки) в массив типа OLEVariant:
Function GetPoint(X, Y, Z: double): OleVariant;

begin

  Result:= VarArrayCreate([0, 2], varDouble);

  Result[0]:= x;

  Result[1]:= y;

  Result[2]:= z;

end;

Попробуем нарисовать окружность. Для этого введем переменную C типа IAcadCircle:
VAR c:IAcadCircle;
…
// зеленая окружность с центром в (20;20;20) радиусом 10
c:=AcadDocument1.ModelSpace.AddCircle(GetPoint(20,20,0),10);

c.color:=acGreen;

c.Lineweight:=acLnWt040;
Аналогичным образом можно подключаться к таким программам, как Visio, SolidWorks, Компас, что позволяет использовать их как основу для разработки специализированных САПР. 

